

# AMSTRAD

PROGRAMANDO MI

AND 1 Núm. 1 - 450 Ptas. (I.V.A. Incluido)




BARROTES II

CHLOAMCHLA

PANTALL

CREACION  
DE  
CARACTERES



R. CARRALÓN

# AMSOFT Serie Oro

La mejor selección de juegos para AMSTRAD

## CAMPEONES DEL MUNDO DE RALLYES



Ponte al volante de tu bólido y ¡a correr! Participan hasta ocho corredores en una misma carrera, que consta de seis etapas, que habrás de recorrer en un tiempo mínimo.  
CASSETTE Y DISCO.

## RAID



¡Detén un ataque nuclear asaltando el Centro de Defensa Soviético! Un juego de acción de múltiples pantallas y diferentes niveles de destreza.  
DISCO.

## SORCERY PLUS



Lucha en busca de los Sorcerers. Sólo si liberas a todos podrás derrotar a los Necromancers. Encontrarás objetos que te servirán de ayuda o confusión. Descubre los pasadizos secretos, y disfruta de uno de los mejores juegos de aventuras de todos los tiempos.  
DISCO.

## FUTBOL



Disfruta la emoción de uno de los deportes más populares del mundo. Juega contra el ordenador (jugador fuerte), o contra tus amigos, quizá, más fáciles de vencer.  
CASSETTE Y DISCO.

## 3D GRAND PRIX



Compite en una carrera de Fórmula 1, en uno de los 8 circuitos internacionales. Guía tu prototipo, acelerando, frenando y cambiando de marchas, mientras tus competidores te pisan los talones.  
CASSETTE Y DISCO.

## AJEDREZ TRIDIMENSIONAL



Para jugadores de cualquier nivel. Proporciona numerosas posibilidades: repetición de movimientos, ver la partida desde el principio, análisis de posiciones, estudio del desarrollo completo de una partida, tablero tridimensional y convencional, etc...  
CASSETTE Y DISCO.

## 3D BOXING



Participa en uno de los deportes más extenuantes gracias a esta magnífica simulación gráfica tridimensional. Entrena a los mejores pugiles: MAD JOE, QUASI Y ROLAND.  
CASSETTE Y DISCO.

## SUPERTRIPPER



Tú, Supertripper, has de buscar los 28 disquetes desperdigados por el planeta Khuh, y salir de allí. Con ayuda de los globos escapa de los aborígenes que te debilitan en los encontronazos.  
CASSETTE Y DISCO.

P.V.P.

CASSETTE

2.300 pts.;

DISCO

3.000 pts.

**AMSTRAD** ESPAÑA

Avda. del Mediterráneo, 9. Tels. 433 45 48 - 433 48 76. 28007 MADRID  
Delegación Cataluña: Tarragona, 110 - Tel. 325 10 58. 08015 BARCELONA

# AMSTRAD

## EDITORIAL

Editorial Cometa, S. A.  
C/. Zamora, s/n. Nave 11.  
San Fernando de Henares  
(MADRID)

## DIRECCION TECNICA

Jesús Negrete Vaquero

## COLABORADORES

Eugenio Garrido.  
Francisco Blázquez.  
Nieves Vera.  
Javier Garrido.  
Firmo Zurdo.

## SECRETARIA DE REDACCION

Pilar Barrio.

## PUBLICIDAD

Edited  
C/. Colombia, 41. 1.º B  
Telf. 458 81 34

## REDACCION Y ADMINISTRACION

C/. Zamora, s/n. Nave 11  
San Fernando de Henares  
(MADRID)

## FOTOCOMPOSICION


Herrata, S. A.  
C/. Alejandro González, 7  
28028 MADRID

## IMPRIME


Gráficas Proa  
D. L.: M-5818-1986

## DISTRIBUYE


Coedis  
C/. Valencia, 245  
Barcelona.




Programa de Dibujo que te permitirá realizar todo tipo de gráficos a color, que tu imaginación pueda realizar.



Este programa te permitirá retroceder en el tiempo y reencarnarte en un fabuloso héroe de la mitología griega.



Programa ideado para estudiar sobre tablas de valores.



Con este artículo podrás aprender a definir los caracteres, para poder emplearlos en tus juegos.

**D**e un modo sencillo y breve, vamos a tratar de explicar el método para crear y construir nuestros propios caracteres.

Tanto el Amstrad CPC-464 y el CPC-664 tienen un total de 256 caracteres definidos, de los cuales, los 32 primeros (incluido el 0) tienen una función de control y que ahora no vamos a analizar.

Por ello, a la hora de definir, vamos a olvidarnos de estos y usar del 32 al 255, ambos inclusive, con los que tenemos suficiente.

En el ejemplo n.º 1 de la cinta podrás ver en pantalla estos caracteres (del 32 al 255).

Si has mirado el manual, habrás observado que todo carácter va definido dentro de una cuadrícula de 8x8, donde los cuadros ocupados por dicho carácter, van pintados y el resto, vacíos.

Basándote en esto, en una hoja de papel cuadriculado, sobre una cuadrícula de 8x8, crea tus propios caracteres. Mira el ejemplo 2.

Una vez obtenido el carácter deseado, ¿cómo introducirlo al ordenador? Para ello, el BASIC de AMSTRAD cuenta con dos comandos:

— SYMBOL AFTER (variable numérica).

— SYMBOL N, a1, a2, a3, a4, a5, a6, a7, a8.

Es fundamental que domines estas dos funciones, y por eso, vamos a explicarlas:

**SIMBOL AFTER X** (Es X una variable numérica)

Este comando nos sirve para «avisar» al ordenador, desde qué número de carácter vamos a redefinir. El valor de X va desde 0 a 255.

Por ejemplo, si yo usara el comando «Symbol after 236», podría redefinir un total de 20 caracteres, es decir, del 236, inclusive, hasta el 255. No es forzoso que yo defina 20 caracteres, si no que como mucho, tras la orden anterior, sólo podría definir 20, pudiendo usar, de esos 20, sólo uno, o dos, o tres, etc.

Este comando no es de uso obligatorio. Si lo omites a la hora de hacer tus programas, el ordenador te pone el

**E**n primer lugar y para una mayor comprensión de lo que a continuación se va a explicar, sería conveniente que cargaras en tu ordenador el programa «EJEMPLOS».

# CREACION DE CARACTERES

CINTRA 1PB  
DISCO "TEXT" 4K ACB  
"TEXT INFO" 7K ACB

límite de redefinición en 240, quedando para tu definición sólo 16 caracteres.



**SYMBOL n, a1, a2, a3, a4, a5, a6, a7, a8**

La variable **n** indica al ordenador el número de carácter que estás redefiniendo. El resto de las variables (al ... a8), son datos que el ordenador usará para rellenar la cuadrícula del modo que tú quieras.


Estos datos se pueden obtener por dos métodos:

- En binario.
- En hexadecimal.

**EN BINARIO.** Haciendo uso de la notación binaria (0-1), de modo que el 0 represente a los cuadrados vacíos y



**DIAGRAMA DE BLOQUES  
DEL PROGRAMA EJEMPLOS**



# CREACION DE CARACTERES

el 1 a los rellenos, podemos «escribir por filas» (en binario) toda la cuadrícula con el carácter incluido, (ejemplo 3).

En ese orden, siendo la fila superior, a2 la siguiente, etc; es como deberemos introducirlo en el ordenador.

Para ello, tras poner Symbol y el número de carácter que estamos redefiniendo (n), hemos de introducir en el orden ya citado y separados por comas, el resto de los datos, anteponiendo a cada fila el symbol «&X», que indica al ordenador que vamos a usar una notación en binario, (ejemplo 4).

Ahora, habiendo visto el ejemplo, asignad a vuestro carácter un número y escribirlo en forma binaria, sin olvidar anteponer «&X».

Es fácil, con un poco de práctica conseguirás mucha rapidez.

**EN HEXADECIMAL.** El sistema de relleno de cuadrícula es el mismo

que en binario. La diferencia es que ahora, en vez de asociar a a1 el valor &X00011110 (por ejemplo), tan solo deberemos de asociar una variable numérica. Y así el resto de filas. Pero ¿cómo conseguir esta variable numérica?

En primer lugar tomemos los datos en binario de nuestro carácter. Veamos el ejemplo 5.

La diferencia con el ejemplo 4 radica en que han aparecido unas cifras que dan valor a cada una de las columnas de la cuadrícula.

Estos valores están así establecidos y son los que nos van a permitir hallar el valor hexadecimal de nuestros datos binarios.

Veamos el ejemplo 6. Junto a la escritura binaria de la cuadrícula, ha aparecido el valor hexadecimal correspondiente a cada fila para el carácter 240.

¿Cómo se han hallado?

```
10 '***** EJEMPLOS *****          ***** DE AYUDA A LA REVISTA *****
    *** Fco.J.Blazquez - ??? (c) - 85 *          *****
*****
20 MODE 1:INK 1,26:INK 0,0:BORDER 0:INK 2,2:INK 3,18
30 GOTO 590
40 CLS:'***** EJEMPLO 1 *****
50 PRINT:PRINT:PRINT "   JUEGO DE CARACTERES REDEFINIBLES
-----":PRINT:PRINT:PEN #2,3
60 WINDOW #2,2,39,6,25
70 FOR n=32 TO 255
80 PRINT #2," ";CHR$(n);" ";
90 SOUND 1,n*2:NEXT
100 RETURN
110 CLS:'***** EJEMPLO 2
120 PEN 1:LOCATE 20,7:PRINT CHR$(143);CHR$(143):LOCATE 19,8:PRINT STRING$(4,CHR$(143)):LOCATE 18,9:PRINT STRING$(6,CHR$(143)):LOCATE 17,10:PRINT STRING$(8,CHR$(143))
130 FOR n=11 TO 14:LOCATE 20,n:PRINT CHR$(143);CHR$(143):NEXT
140 FOR m=0 TO 128 STEP 16:ORIGIN 254+m,175:DRAW 0,128,2:ORIGIN 254,175+m:DRAW 128,0,2:NEXT
150 PEN 3:LOCATE 12,4:PRINT "CARACTER n: 240"
160 LOCATE 27,11:PEN 1:PRINT "=" " ";CHR$(240)
170 RETURN
180 CLS:'***** EJEMPLO 3
190 PEN 1
200 PEN 1:LOCATE 10,7:PRINT STRING$(2,CHR$(143)):LOCATE 9,8:PRINT STRING$(4,CHR$(143)):LOCATE 8,9:PRINT STRING$(6,CHR$(143)):LOCATE 7,10:PRINT STRING$(8,CHR$(143))
210 FOR n=11 TO 14:LOCATE 10,n:PRINT CHR$(143);CHR$(143):NEXT
220 PEN 3:LOCATE 2,4:PRINT " CARACTER n: 240          NOTACION BINARIA
-----"
230 FOR m=0 TO 128 STEP 16:ORIGIN 94+m,175:DRAW 0,128,2:ORIGIN 94,175+m:DRAW 128,0,2:NEXT
240 PEN 1:FOR l=7 TO 14:LOCATE 18,l:PRINT "=" :NEXT
250 WINDOW #1,23,30,7,15:PRINT #1,"00011000001111000111110111111110001100000011000001100000011000"
260 IF papo=4 THEN GOTO 280
270 RETURN
280 '***** EJEMPLO 4
290 PEN 2:FOR l=7 TO 14:LOCATE 21,l:PRINT "&X":NEXT
300 RETURN
310 CLS:' ***** EJEMPLO 5
```

Pues sumando los valores de los cuadrados rellenos de cada fila, siguiendo los valores antes mencionados.

Practica tu ahora traspasando tus caracteres en binario a hexadecimal y viceversa. No olvides que en hexadecimal no has de poner el símbolo «&X» antes de cada dato.

El resultado de ambos métodos es el mismo, aunque con la práctica te será más cómodo utilizar el hexadecimal pues no sólo es más rápido sino que ahorra memoria a la hora de almacenar datos.


Bueno, pues ahora que ya sabéis redefinir caracteres mirad el ejemplo n.º 7. En él tu podrás ver en pantalla tus caracteres recién creados.

Y por último, y para que la «lección» quede completa, sería aconsejable que cuando hayas comprendido bien esta explicación, cargarás en tu ordenador el programa «TEST». En

él se te cuestionará acerca de conceptos que se han mencionado y que, sin duda, te será muy útil el recordarlos a la hora de asimilar todos tus nuevos conocimientos.

Y para que el programa n.º 7 te sea más útil aún, te proporcionamos ejemplo de caracteres redefinidos en hexadecimal, los cuales podrás incorporar como rutinas a tus programas. No te decimos de que se trata cada uno para mantenerte en la incognita.

- carácter 1** 129,255,60,126,195,219,126,231
- carácter 2** 126,219,255,195,90,90,126,231
- carácter 3** 60,102,66,231,189,189,36,231
- carácter 4** 24,60,36,60,102,102,255,231
- carácter 5** 129,153,255,90,255,195,90,255



```

320 PEN 1:LOCATE 5,7:PRINT STRING$(2,CHR$(143)):LOCATE 4,8:PRINT STRING$(4,CHR$(
143)):LOCATE 3,9:PRINT STRING$(6,CHR$(143)):LOCATE 2
,10:PRINT STRING$(8,CHR$(143))
330 FOR l=7 TO 14:LOCATE 5,1:PRINT CHR$(143);CHR$(143):NEXT
340 LOCATE 1,2:PEN 1:PRINT " 128
8
350 LOCATE 1,17:PRINT " 64
4
360 FOR m=0 TO 128 STEP 16:ORIGIN 14+m,175:DRAW 0,128,2:ORIGIN 14,175+m:DRAW 128
,0,2:NEXT
370 ORIGIN 37,174:DRAW 0,-25,3:ORIGIN 73,174:DRAW 0,-41:ORIGIN 105,174:DRAW 0,-5
6:ORIGIN 135,174:DRAW -10,-70
380 ORIGIN 22,305:DRAW 10,60,3:ORIGIN 55,305:DRAW 0,45:ORIGIN 87,305:DRAW 0,30:0
RIGIN 119,305:DRAW 0,14
390 ORIGIN 38,174:DRAW 0,-25:ORIGIN 71,174:DRAW 3,-41:ORIGIN 100,174:DRAW 6,-57:
ORIGIN 132,174:DRAW -8,-73
400 RETURN
410 ***** EJEMPLO 6
420 WINDOW #4,13,22,7,15:PRINT #4,"&X00011000&X00111100&X01111110&X11111111&X00
11000&X00011000&X00011000&X00011000"
430 PEN 3:LOCATE 10,4:PRINT "NOTACION BINARIA HEXADECIMAL"
440 WINDOW #3,33,35,7,15:PRINT #3," 24 60128255 24 24 24 24"
450 RETURN
460 ***** EJEMPLO 7
470 CLS:PEN 1:PRINT:PRINT:PRINT " ESTE PROGRAMA EXPONDRA EN PANTALLA EL C
ARACTER QUE HAYAS CREADO "
480 PRINT:PRINT " EN CASO DE INTRODUCIR LOS DATOS EN BINARIO,NO HAS DE OL
VIDARTE DE PONER < &X > DELANTE DE LAS OCHO CIFRA
S EJ: &X11100111"
490 PRINT:PRINT:PRINT " EN CASO DE ROMPERSE EL PROGRAMA ,SE DEBERA A UNA INT
RODUCCION INCORRECTA DE ALGUN DATO: TECLEA < RU
N > Y VUELVE A INTENTARLO CON CUIDADO"
500 SYMBOL AFTER 230
510 FOR x=1 TO 8
520 PRINT:PRINT " El valor de a";x;:INPUT " es ";a(x)
530 IF a(x)<0 OR a(x)>255 THEN PRINT:PRINT " ESE DATO ES INCORRECTO:CORRIGELO"
:LET x=x-1
540 NEXT
550 SYMBOL 231,a(1),a(2),a(3),a(4),a(5),a(6),a(7),a(8)
560 CLS:LOCATE 1,6:PRINT:PRINT:PRINT " ESTE ES TU CARACTER = ";CHR$(231)

```

# CREACION DE CARACTERES

## RUTINA PARA EXPRESARTE CORRECTAMENTE EN ESPAÑOL

Esta rutina es muy util para programas en los que debas escribir textos en ESPAÑOL. Cuando la adaptes a cualquier programa renumérala a tu gusto.

Es la definición de los caracteres Ñ, ñ, í y, ÷.

Carácter	Tecla donde aparece
Ñ	[+SHIFT
ñ	[
í	]
÷	] +SHIFT

10 SYMBOL AFTER 32  
20 SYMBOL  
125,0,24,0,24,48,46,102,60  
30 SYMBOL  
93,24,0,24,24,24,24,24,24  
40 SYMBOL  
91,60,0,216,102,102,102,102,0  
50 SYMBOL  
123,216,230,246,222,206,198,198,0

Tras teclear el programa y ejecutarlo, te dará el mensaje «ready» en pantalla. Esto significa que no has cometido ningún error de copiado y que los caracteres ya han sido redefinidos. Prueba las teclas que antes hemos citado y comprueba lo util que resulta esta rutina. ■

## 5 REM: PROGRAMA DE ARREGLO DEL ESPAÑOL

```

570 FOR m=1 TO 20 STEP 2:LOCATE 10+m,13:PRINT CHR*(231):NEXT
580 RETURN
590 BORDER 0:CLS:PEN 3:PRINT "          ILUSTRACIONES
      AAAAAAAAAAAAAA"
600 LOCATE 1,24:PEN 2:PRINT " FCO.J.BLAZQUEZ (c) ED.COMETA -1985-":LOCATE 1,5
610 PRINT:PRINT:PRINT:PEN 1:PRINT " Esta seccion ha sido ideada con la final
idad de servir como complemento ilustrativo a los
conceptos que en la revista se explican."
620 PRINT:PRINT " Escoja en cada momento el ejemplo que en ella se mencione..
.":PRINT
630 FOR v=1 TO 7:PRINT " i";v;"2.... Ejemplo";v:NEXT
640 GOSUB 850:FOR n=1 TO 20:READ h:SOUND 1,h,20:SOUND 2,h,18,14:SOUND 3,h-(h/2),
20,13
650 IF INKEY(64)=0 THEN GOSUB 730
660 IF INKEY(65)=0 THEN GOSUB 740
670 IF INKEY(57)=0 THEN GOSUB 750
680 IF INKEY(56)=0 THEN GOSUB 760
690 IF INKEY(49)=0 THEN GOSUB 770
700 IF INKEY(48)=0 THEN GOSUB 780
710 IF INKEY(41)=0 THEN GOSUB 790
720 NEXT:GOTO 640
730 GOSUB 40:GOTO 800
740 GOSUB 110:GOTO 800
750 GOSUB 180:GOTO 800
760 GOSUB 180:GOSUB 280:GOTO 800
770 GOSUB 310:GOTO 800
780 GOSUB 310:GOSUB 410:GOTO 800
790 GOSUB 460:GOTO 800
800 PEN 1:LOCATE 6,24:PRINT " PULSA TECLA ¡M¿ PARA MENU "
810 FOR n=1 TO 3 STEP 0.08:LOCATE -4+(n/0.09),2:PEN n:BORDER n*3:PRINT " ";CHR*(
248+n);" ":LOCATE 40-(n/0.09),25:PRINT " ";CHR*(24
0+n);" ":SOUND 1,n*50,10:IF INKEY(38)=0 THEN GOTO 590
820 NEXT
830 GOTO 810
840 '***** MELODIA MENU *****
850 DATA 638,478,478,478,638,506,506,506,638,478,478,478,402,478,478,402,402,506
,506,506
860 RESTORE 850:RETURN
870 FOR n=1 TO 3 STEP 0.08:LOCATE -2+(n/0.09),20:PEN n:PRINT " ";CHR*(143);"
":SOUND 1,n*50,10:NEXT:GOTO 870

```

## Uso del BREAK

**E**sta tecla se usa para cambiar el color de cualquiera de los lápices, fondo o borde actuales. En primer lugar y tras haber pulsado dos veces BREAK, comienza la selección del color del borde. Para variar y probar usa la tecla COPY. Cuando hayas encontrado el color idóneo, pulsa ENTER para fijarlo. Tras ello, se procede a operar igual con el fondo. Seleccionado y fijando del mismo modo. De similar manera y por el orden siguiente, podrás seleccionar tras el fondo:

La tinta 1

La tinta 2


La tinta 3

Habiendo fijado la tinta 3, el programa vuelve a la rutina principal.

Cuando estés cambiando la tinta, observarás que aparece un marco rodeando el fondo. Este tiene como misión informarte del color que estás probando en caso de no haber pintado nada con esa tinta.

### Ventana información

Te pide que tipo de información requieres. He aquí una lista de los números asignados a cada campo de información.



# PANTALL

## Líneas Principales

Línea	FUNCION
100	— Llamada a la pantalla de instrucciones.
110-150	— Fija valores de las variables.
160-570	— Bucle principal. Mueve posiciones del cursor y analiza la posición de todas las teclas usadas del programa. Si alguna está pulsada entonces se desvía hacia la rutina que corresponda. Controla y corrige los valores de la tinta, la velocidad del cursor, el grado inicial, el tamaño de la raya, el grado final, el STEP para representaciones de curvas y la longitud de los radios uno y dos.

*Pasa a página 10*

1. Velocidad del cursor.
2. Longitud radio 1.
3. Longitud radio 2.
4. STEP de los radios.
5. Grado de origen.
6. Grado final.
7. Longitud de raya.
8. Coordenadas cursor.
9. Coordenadas fijadas.
10. Volver al bucle principal.

Según lo que elijas, te manda a una línea de la rutina que a continuación citamos.

### Informa según lo elegido

Como ya hemos mencionado, proporciona los datos requeridos y retorna a la opción de información.

### Opción cursor

Condiciona el manejo del cursor gráfico por medio de teclas.

### Opción Joystick

Condiciona el cursor gráfico

a ser controlado por medio de Joystick.

## PANTALL

### COMENTARIO

Como habrás podido observar, este es un programa de carácter práctico y didáctico. Su finalidad es poder usar los comandos gráficos del Amstrad de un modo sencillo y fácil.

El programa podía haber sido más amplio, pero hemos creído que con su estructura actual es más que suficiente para comprender el uso de gráficos en este ordenador.

En cualquier caso, al final de este apartado proporcionamos alguna rutina y líneas que podrás añadir al programa, dotándolo así de una mayor complejidad.

Para tu perfecta comprensión, hemos formatea-

do todo el listado por medio de REM que especifican claramente la función de la rutina a que se refieren.

Vamos ahora a examinar y comentar rutina por rutina de este programa.

### Inicialización

Desvía el programa hacia la línea 1800, donde comienzan las instrucciones.

### Variables fijadas

En este apartado, se han dado los valores iniciales a algunas variables, para que no tengas que inicializarlos todas por medio de las teclas. Se definen los colores y las ventanas. El programa se desvía (con inmediata vuelta) a la rutina de manejo del cursor por el teclado.

### Bucle principal

Se analiza la situación de todas las teclas en juego. A través de ellas se regulan todas las variables, como los radios, los colores o los grados.

Además se llama a las diversas rutinas de ejecución según que tecla se tenga pulsada. Se mueve el cursor gráfico y se desvía el programa a la siguiente rutina.

### Cursor visual

Se hace visible el cursor gráfico. Tras ello, se retorna el bucle principal.

### Elipses o círculos vacíos

Con esta fórmula, el ordenador ejecuta a partir de dos ejes una proyección de radios con la sentencia PLOT. Con ello se consiguen los espacios vacíos.

Para lograr una circunferencia, iguala los dos radios.

### Elipses o círculos llenos

Si te fijas la fórmula es igual que la anterior, tan

viene de página 9

Línea	FUNCIÓN
580-660	— Hace visible el cursor gráfico. Además de esto, hay una fórmula inserta en esta rutina que hace posible que el cursor no borre las figuras cuando pase por encima de ellas.
670-680	— Hace elipses vacías o circunferencias dependiendo de los radios. Para ello usa los valores de gi, gf, st, r1, r2, x, y, l.
690-700	— Dibuja elipses o círculos llenos dependiendo de los radios. Para ello usa los valores de las variables anteriores.
710-720	— Hace cuadrados a partir del uso del DRAW (coordenadas relativas) y con el valor de da como lado.
730-740	— Fija las coordenadas fijas.
750-760	— Dibuja una raya entre las coordenadas fijas y el cursor. Se usa el PEN l.
770-780	— Traza una línea de longitud da y de forma horizontal, con la posición del cursor como centro.
790-800	— Traza una línea de longitud da y de forma oblicua hacia la derecha con la posición del cursor como centro.
810-820	— Traza una línea de longitud da y de forma horizontal vertical con la posición del cursor como centro.
830-840	— Traza una línea de longitud da y de forma oblicua hacia la izquierda.
850-860	— Limpia la pantalla. Uso de CLS.
870-880	— Guarda en una cinta o disco la pantalla elegida. Para ello, es preciso que antes introduzcas un nombre.
890-910	— Carga un dibujo desde la cinta o el disco a pantalla. Como antes, debes introducir el nombre antes.
920-1230	— Cambia los colores del fondo, borde y tinta. Su distribución es la siguiente 930-980 color borde 990-1040 color fondo 1050-1100 tinta 1 1110-1160 tinta 2 1170-1230 tinta 3
1240-1360	— Te pide qué concepto de información quieres y según la respuesta se desvía a la rutina adecuada:

pasa a página 11

solo con la variación del comando DRAW en vez de PLOT.

### Cuadrado

Con el cursor como centro, hace cuadrados de lado da, la variable que se regula con las teclas 7-8.

### Fija coordenadas x, y

Da el valor actual de x e y a x1 e y1 respectivamente. Por tanto, hasta que no vuelva a usar esta rutina, las coordenadas fijas no varían.

### Dibuja línea de coordenadas fijas

Examina coordenadas actuales del cursor, las resta con las fijadas, y traza una línea de unas a otras.

### Traza línea (-), (I), ( ), (/)

Con el cursor como centro, se traza una línea de la forma que corresponda, cuyo tamaño viene regulado por las teclas 7-8.

### Limpia pantalla

Tan sólo por medio del comando CLS; BORRA todo lo que haya en pantalla.

### Save pantalla

Pide el nombre en la ventana de información y tras ello «vuelca» el contenido de la pantalla en cinta o disco.

### Load pantalla

Después de haber introducido el nombre del dibujo requerido, pasa a cargarlo en pantalla.

### Instrucciones

En esta rutina se explican generalidades del juego, así como las instrucciones de manejo. Tras ello se pasa a la posibilidad de uso de disco.

### Opción disco

Tan sólo funciona con ordenadores dotados de unidad de disco. En caso de ser elegida con el CPC-464, dará error. Teclea «RUN» y inicializa el juego.

### \*NOTAS\*

— Para borrar utiliza la tinta 0 y pinta por encima.

— Apunta el nombre de los dibujos grabados para no olvidarte de ellos a la hora de recuperarlos.

— Si el cursor se sale de pantalla y no hay forma de localizarlo, usa la ventana de información pidiendo sus coordenadas. A partir de ahí te será más fácil encontrarlo.

— Si quieres que la grabación, y por tanto la posterior carga, sea más rápida, antes de iniciar el juego rómpelo con el BREAK cuando estés en las instrucciones. Entonces teclea: SPEED WRITE 1 y luego RUN.

Para finalizar te diremos que puedes ampliar mucho las posibilidades de este programa poniéndole otras funciones. Por ejemplo, si tu ordenador tiene unidad de disco incorporada, trata de añadirle el comando FILL.

El sistema a seguir para cualquier aumento sería el siguiente.

1.º Introduce dentro del bucle principal un «IF inkey (x)=0 then», donde x tenga el valor de la tecla que le asignes a tal función.

2.º Tras ello, coloca un «GOSUB H» detrás de la condicional (en la misma línea) siendo h no una variable sino un valor de línea.

3.º En ese valor por ti asignado, comienza la rutina de ejecución de esa función.

*viene de página 10*

Línea	FUNCION
1370-1470	— Informa sobre el campo de información por ti elegido. Son 10 rutinas en total.
1480-1530	— Da valores a la variable jok y siguientes para poder mover el cursor a través del cursor del teclado.
1540-1590	— Al igual que el anterior, selecciona el modo de control del cursor. Ahora es a través del joystick.
1600-1930	— Pantalla de presentación e instrucciones. Tras todo ello opción a usar o no el disco. Caso de ser respuesta afirmativa, manda a rutina disco.
1940-1950	— Opción disco. En caso de ser seleccionada con ordenadores sin disco, dará error.

### Variables Principales

VARIABLE	FUNCION
x, y	— Actual posición del cursor de gráficos.
da	— Es la longitud de la raya usada con las teclas D, R, E, F. También es el lado de los cuadrados.
st	— Es el STEP en funciones curvas.
vc	— Aunque llamada velocidad del cursor, esta variable equivale a el número de unidades que aumenta o disminuye la X o la Y en cada movimiento.
r1	— Es el tamaño del radio 1 en las funciones curvas.
r2	— Es el tamaño del radio 2 en las funciones curvas.
gi	— Este programa te ofrece la particularidad de variar el grado de origen de las representaciones curvas. Ten cuidado de que no sea mayor que el grado final.
gf	— Grado final. En él acaba la función curva. Si la diferencia entre gf y gi es igual o mayor que 360°, entonces la curva se cerrará.
l	— Color que se usa en cada momento (Tinta)
jok	— Esta variable toma distintos valores según se elija el cursor o el joystick para controlar el cursor.

*pasa a página 12*

# PANTALL

# PANTALL

viene de página 11

VARIABLE	FUNCION
jok(1)...jok(4)	— Identica función que la anterior.
nombres\$	— Nombre asignado a una pantalla para cargar o grabar.
retardo	— Como su propio nombre indica, funciona como pausa en la ejecución.
ib	— Número de color asignado a la tinta del borde mediante la tecla COPY.
i0	— Número de color asignado a la tinta del fondo mediante la tecla COPY.
i1	— Número de color asignado a la tinta del lápiz 1 mediante la tecla COPY.
i2	— Número de color asignado a la tinta del lápiz 2 mediante la tecla COPY.
i3	— Número de color asignado a la tinta del lápiz 3 mediante la tecla COPY.
IS	— Variable que recoge que tipo de información quieres obtener.
dci\$	— Opción disco.
9*	— Tinta 0 (Pen 0)
6*	— Tinta 1 (Pen 1)
3*	— Tinta 2 (Pen 2)
Enter*	— Tinta 3 (Pen 3)
7-8	— -/+ Velocidad o «salto» del cursor gráfico.
1-2*	— -/+ Grados de origen para representación de curvas.
0-.*	— -/+ Grados de fin para representación de curvas.
i-[	— -/+ radio 1.
+] ]	— -/+ radio 2.
/-	— -/+ «STEP» en representaciones de curvas.
A	— Dibuja círculos o elipses llenas, según sean o no los radios iguales
Q	— Dibuja circunferencias o elipses vacías, según sean o no los radios iguales
Z	— Cuadrados (el tamaño de los lados se regula con las teclas 7-8)
W	— Hace fija una coordenada, la actual del cursor.
X	— Traza una línea desde el cursor hasta la coordenada últimamente fijada
E	— Traza una línea horizontal de longitud regulada por las teclas 7-8.

pasa a página 13

4.º Inmediatamente después, numera esta línea con un RETURN para que vuelva al bucle inicial tras haber realizado la función.

Para un mejor entendimiento, te proporcionamos a continuación dos ejemplos de lo anterior.

## \*RUTINA DE LAPIZ FINO\*

Con esta rutina, cada vez que pulse la tecla B se imprimirá 1 punto de pantalla. Si quieres hacer una línea, coloca la velocidad del cursor en 1 y pulsa B a la vez de mover el cursor a tu gusto.

```
175 IF INKEY (54) = 0
THEN GOSUB 2050
2050 REM: ** ROMBOS
**
```

```
2060 origin x-da, y: draw
da, da, l: drawr da,
-da:drawr -da, -da: drawr
-da, da
2070 RETURN
```

```
175 IF INKEY (54) = 0
THEN GOSUB 2000
2000 REM: ** LAPIZ
FINO **
2010 ORIGIN x,y:PLOT
0,0,1
2020 RETURN
```

## \*RUTINA DE ROMBOS\*

Al pulsar la tecla G, aparecerá en pantalla y con el cursor como centro, un rombo de lados iguales, de tamaño da. Aumenta y disminuye el tamaño con teclas 7 y 8.

```
176 IF INKEY (52) = 0
THEN GOSUB 2050
2050 REM: **ROMBOS**
2060 ORIGIN x-da,y:
DRAWR da, da, l:
DRAWR da,-da: DRAWR
-da,-da: DRAWR -da,da
2070 RETURN
```

viene de página 12

VARIABLE	FUNCION
R	— Traza línea oblicua hacia la derecha de longitud regulada por las teclas 7-8.
D	— Traza línea vertical de longitud regulada por teclas 7-8.
F	— Traza línea oblicua hacia la izquierda de longitud regulada por teclas 7-8.
C	— Limpia pantalla.
S	— Save pantalla o cinta o a disco, según se elija.
L	— Load pantalla de cinta o de disco, según se elija.
Break	— Se usa para cambiar los colores en uso. Dando al COPY se prueba el color de cada apartado, siendo el borde el primero, el fondo el 2.º, y la tintas por último. Una vez elegido el color de cada apartado, se fija el mismo y se pasa al siguiente apartado pulsando ENTER.

VARIABLE	FUNCION
Copy	— Ver uso de Break.
Enter	— Ver uso de Break.
J	— Mueve el cursor gráfico con joystick.
K	— Mueve el cursor gráfico con el cursor, sobre el teclado numérico.
I	— Da información requerida. Pregunta de que informar. Respondiendo las claves siguientes: «1» Da la velocidad del cursor. «2» Longitud radio 1. «3» Longitud radio 2. «4» STEP para curvas. «5» Grado inicial u origen. «6» Grado final. «7» Longitud raya. «8» Coordenadas actuales del cursor. «9» Ultimas coordenadas fijadas. «10» Abandonar la opción de información.

\*Del teclado numérico.


```
10 '*****
20 '*****
30 '**
40 '**          PANTALL          **
50 '**
60 '**          Fcº.J.BLAZQUEZ 1985  **
70 '**
80 '*****
90 '----- INICIALIZACION -----
100 GOTO 1800
110 '----- VARIABLES FIJADAS -----
120 x=200:y=200:l=1
130 da=20:st=1:vc=2:r1=100:r2=110:gi=1:gf=360
140 MODE 1:INK 0,0:BORDER 0:INK 1,24:INK 2,18:INK 3,6
150 WINDOW R3,1,1,25,25:WINDOW R4,1,39,25,GOSUB 1480
160 '***** BUCLE PRINCIPAL *****
170 IF INKEY(37)=0 THEN GOSUB 1490
180 IF INKEY(45)=0 THEN GOSUB 1550
190 IF INKEY(3)=0 THEN l=0
200 IF INKEY(4)=0 THEN l=1
210 IF INKEY(5)=0 THEN l=2
220 IF INKEY(6)=0 THEN l=3
230 IF INKEY(10)=0 THEN da=da-1
240 IF INKEY(11)=0 THEN da=da+1
250 IF INKEY(20)=0 THEN vc=vc-1
260 IF INKEY(12)=0 THEN vc=vc+1
270 IF INKEY(13)=0 THEN gi=gi-1
280 IF INKEY(14)=0 THEN gi=gi+1
```

# PANTALL

```
290 IF INKEY(15)=0 THEN gf=gf-1
300 IF INKEY(7)=0 THEN gf=gf+1
310 IF INKEY(30)=0 THEN st=st-1
320 IF st<1 THEN st=1
330 IF INKEY(22)=0 THEN st=st+1
340 IF INKEY(26)=0 THEN r1=r1-1
350 IF INKEY(17)=0 THEN r1=r1+1
360 IF INKEY(28)=0 THEN r2=r2-1
370 IF INKEY(19)=0 THEN r2=r2+1
380 IF INKEY(69)=0 THEN GOSUB 700
390 IF INKEY(67)=0 THEN GOSUB 670
400 IF INKEY(71)=0 THEN GOSUB 720
410 IF INKEY(59)=0 THEN GOSUB 740
420 IF INKEY(63)=0 THEN GOSUB 760
430 IF INKEY(60)=0 THEN GOSUB 870
440 IF INKEY(36)=0 THEN GOSUB 900
450 IF INKEY(58)=0 THEN GOSUB 780
460 IF INKEY(61)=0 THEN GOSUB 800
470 IF INKEY(50)=0 THEN GOSUB 820
480 IF INKEY(53)=0 THEN GOSUB 840
490 IF INKEY(jok)=0 THEN y=y+vc
500 IF INKEY(jok2)=0 THEN y=y-vc
510 IF INKEY(jok3)=0 THEN x=x-vc
520 IF INKEY(jok4)=0 THEN x=x+vc
530 IF INKEY(62)=0 THEN GOSUB 860
540 IF INKEY(35)=0 THEN GOSUB 1250
550 ON BREAK GOSUB 920
560 GOSUB 590
570 GOTO 160
580 '----- CURSOR VISUAL -----
590 PRINT R3,CHR$(23)
600 ORIGIN x,y
610 MOVER -8,0 : DRAWR 16,0,1
620 MOVER -8,-8: DRAWR 0,16
630 FOR j=0 TO 10:NEXT
640 DRAWR 0,-16: MOVER 8,8
650 DRAWR -16,0: MOVER 8,0
660 PRINT R3,CHR$(23);CHR$(0):RETURN
670 '- Hacer elipses vacias o circunferencias dependiendo de los radios -----
680 FOR g=gi TO gf STEP st:DEG:ORIGIN x,y:PLOT r1*SIN(g),r2*COS(g),1:NEXT:RETURN

690 '- Hacer elipses llenas o circulos dependiendo de los radios -----
700 FOR g=gi TO gf STEP st:DEG:ORIGIN x,y:DRAW r1*SIN(g),r2*COS(g),1:NEXT:RETURN
710 '----- Hacer cuadrado -----
720 ORIGIN x-(da/2),y-(da/2):DRAWR 0,da,1:DRAWR da,0:DRAWR 0,-da:DRAWR -da,0:RET
URN
730 '----- Fija coordenadas x,y -----
740 LET x1=x:LET y1=y:RETURN
750 '- dibuja raya de coordenadas fijas hasta el cursor -----
760 ORIGIN x1,y1:DRAW x-x1,y-y1,1:RETURN
770 ' --- Traza linea (-) -----
```

*pasa a página 16*



«Pantall»

FIGURA I

# PANTALL

viene de página 14

```
780 ORIGIN x-(da/2),y: DRAW da,0,1: RETURN
790 ' --- Traza linea / ----
800 ORIGIN x,y-(da/2): DRAW 0,da,1: RETURN
810 ' --- Traza linea ñ ---
820 ORIGIN x-(da/2),y-(da/2): DRAW da,da,1: RETURN
830 ' --- Traza linea ñ ---
840 ORIGIN x-(da/2),y+(da/2): DRAW da,-da,1: RETURN
850 '----- Limpia la pantalla -----
860 CLS: RETURN
870 '----- SAVE pantalla -----
880 PAPER R4,0: PEN R4,1: INPUT R4," Dame nombre "; nombre$: LET nombre$=LEFT$(nombre$,6): IF nombre$="" THEN GOTO 870
885 LET nombre$="!" + nombre$: CLS R4: SAVE nombre$,B,&C000,&4000
886 RETURN
890 '----- LOAD pantalla -----
900 PAPER R4,0: PEN R4,1: INPUT R4," Dame nombre "; nombre$: LET nombre$=LEFT$(nombre$,6): IF nombre$="" THEN GOTO 890
910 LET nombre$="!" + nombre$: CLS R4: LOAD nombre$,&C000: RETURN
920 '--- HACE USO DEL BREAK PARA VARIAR -
930 '--- EL BORJE,FONDO y TINTAS ----
940 IF INKEY(9)=0 THEN ib=ib+1:
950 FOR m=1 TO 200:NEXT: IF ib>26 THEN ib=0
960 IF ib>26 THEN ib=0
970 IF INKEY(18)=0 THEN GOTO 990
980 BORDER ib: GOTO 940
990 '----- COLOR FONDO -----
1000 IF INKEY(9)=0 THEN i0=i0+1
1010 FOR m=1 TO 200:NEXT: IF i0>26 THEN i0=0
1020 IF i0>26 THEN i0=0
1030 IF INKEY(18)=0 THEN GOTO 1050
1040 INK 0,i0: GOTO 990
1050 '----- TINTA 1 -----
1060 IF INKEY(9)=0 THEN i1=i1+1
1070 FOR m=1 TO 200:NEXT: IF i1>26 THEN i1=0
1080 IF i1>26 THEN i1=0
1090 IF INKEY(18)=0 THEN GOTO 1110
1100 INK 1,i1: ORIGIN 0,0: DRAWR 639,0,1: DRAWR 0,399: DRAWR -639,0: DRAWR 0,-399: GOT 0 1050
1110 '----- TINTA 2 -----
1120 IF INKEY(9)=0 THEN i2=i2+1
1130 FOR m=1 TO 200:NEXT: IF i2>26 THEN i2=0
1140 IF i2>26 THEN i2=0
1150 IF INKEY(18)=0 THEN GOTO 1170
1160 INK 2,i2: ORIGIN 0,0: DRAWR 639,0,2: DRAWR 0,399: DRAWR -639,0: DRAWR 0,-399: GOT 0 1110
1170 '----- TINTA 3 -----
1180 IF INKEY(9)=0 THEN i3=i3+1
1190 FOR m=1 TO 200:NEXT: IF i3>26 THEN i3=0
1200 IF i3>26 THEN i3=0
1210 IF INKEY(18)=0 THEN GOTO 1230
```



# PANTALL

```
1660 PRINT " @-i.....-/+ radio 1 ;-2.....-/+ ra
dio 2 /-ñ.....-/+ Step( curvas )
A.....Circulos o elipses llenas"
1670 PRINT " Q.....Circunferencias o elip
ses vacias Z.....Cuadrados
W.....Fija coordenadas X.....Traza linea desde
curs. a coor.fij."
1680 PRINT:PRINT:PEN 2:LOCATE 1,25:PRINT " PULSA ENTER PARA SEGUIR"
1690 IF INKEY(18)=0 THEN GOTO 1710
1700 GOTO 1690
1710 PRINT:PRINT:PRINT
1720 PEN 1:PRINT " E.....Traza linea - R.....
Traza linea / D.....Traza linea
ñ F.....Traza linea ñ L.....LOAD pantall
a S.....SAVE pantalla"
1730 PRINT " J.....Uso de joystick K.....Uso de
teclado I.....INFORMACION
1740 :PRINT:PRINT " Estas son las claves de cada dato:":PRINT
1750 PRINT " '1'..... Velocidad del cursor '2'..... Longitud radi
o 1 '3'..... Longitud radio 2
'4'..... Step para curvas '5'..... Grado origen"
1760 PRINT " '6'..... Grado final '7'..... Longitud de r
aya '8'..... Coordenadas del cursor
'9'..... Coordenadas fijadas '10'..... Abandonar opcion inform."
1770 PRINT:PRINT:PEN 2:LOCATE 1,25:PRINT " PULSA ENTER PARA SEGUIR"
1780 IF INKEY(18)=0 THEN GOTO 1890
1790 GOTO 1780
1800 INK 0,0:INK 1,26:INK 2,11:INK 3,6:BORDER 0:MODE 1
1810 PEN 3:CLS:PRINT:PRINT " PANTALL
"
1820 PEN 1:PRINT:PRINT:PRINT:PRINT " Con este programa tu podras crear tus
propias pantallas o portadas. Seria con
veniente que leyeras deteni-damente las instrucciones."
1830 PRINT:PRINT " No es tan complicado como a simple vista parece:con un po
co de practica te haras pronto con su manejo."
1840 PRINT:PRINT " En cualquier caso,podras encontrar una explicacion mas detal
lada en la revista que se adjunta con la cinta."
1850 PRINT:PRINT:PEN 2:PRINT " ANIMO ARTISTA!!!"
1860 PEN 3:LOCATE 1,25:PRINT " PULSA ENTER PARA SEGUIR"
1870 IF INKEY(18)=0 THEN GOTO 1600
1880 GOTO 1870
1890 CLS:PRINT:PRINT:PRINT:PEN 1:INPUT " VAS A USAR LA UNIDAD DE DISCO ";dci$
1900 dci$=UPPER$(dci$):PEN 3:LOCATE 1,25:PRINT " Fco.J.BLAZQUEZ (c) ED.COMETA
1985"
1910 IF dci$="S" THEN GOSUB 1950
1920 IF dci$="S" OR dci$="N" THEN LOCATE 1,25:PRINT " Fco.J.BLAZQUEZ (c) ED.CO
META 1985":FOR retardo=1 TO 2000:NEXT:GOTO 120
1930 GOTO 1890
1940 '----- OPCION DISCO -----
1950 LOCATE 1,25:PRINT " Fco.J.BLAZQUEZ (c) ED.COMETA 1985":FOR retardo=1 TO 2
000:NEXT:ñDISC:RETURN
```

**1.** En primer lugar, al redefinir las letras en vez de usar otros símbolos se han usado sus propios caracteres, usurpándoles su propia identidad. Por ello, al listar el programa, habrás observado que muchas palabras conservan esas raras letras que la antigua civilización tenía.

De este modo, te ahorras el redefinir teclas con los comandos KEY y KEY DEF, método que suele usar mucha gente para estos casos.

**2.** En la rutina desvío; si te has fijado al jugar, habrás observado que los marcadores, al pasar de pantalla, no se borra, sino que giran hacia arriba dejando paso a otros nuevos que van apareciendo por debajo, como si fuera un rodillo.

Este efecto se consigue haciendo uso de las ventanas. Analiza esta rutina con detenimiento.

**3.** Dentro del «bloque común» hay varias cosillas que merece la pena detenerse a estudiar.

Aparte de las dos fórmulas claves (el call &BD19, y el test), hay ocho líneas dedicadas al manejo del muñeco. Con 4 bastaría, pero se han puesto 8 para que también se pueda controlar con joystick.

De este modo complaces a todo el mundo a la vez que no por jugar con una opción excluyes la otra. Aparte de esto, acostúmbrate a controlar el movimiento por los comandos INKEY(X)="¿?" (permitiendo también el uso de diagonales).

La posición del call &BD19 ha de ser entre la escritura del muñeco y el borrado del mismo. No es que sea necesario, pero sí que da mejores resultados.

**4.** En las instrucciones: siempre que redefines las letras, pon un espacio entre línea y línea para su mejor lectura.

**A** veces, a la hora de programar, a parte de utilizar las ya mencionadas «fórmulas clave», se hace uso de algunos trucos adquiridos tan sólo con la práctica. En este juego, se ha empleado algún que otro de modo, que para que os sirva de experiencia, vamos a citarlos.

# CHLOAMCHLA

CINTA "1PA  
DISCO KCB "OLIMPIA" 9K



## Comentario

Seguro que alguna vez habéis soñado con las antiguas civilizaciones. A todos nos han admirado esas fabulosas aventuras de los héroes mitológicos. Pues bien, este juego es fruto de un sueño. Para meteros en ambiente voy a contaros la historia.

Mittokropolis era un poblado pequeño de una antigua civilización de la que ya no queda vestigio alguno. Por aquel entonces, Kenthonn, un brujo que habitaba no lejos de allí, robó las 7 joyas mágicas de los Dioses.

Para recuperarlas fue elegido CHLOAMCHLA, un joven del lugar. Los dioses le llevaron hasta los dominios del brujo y le dieron «el oráculo», que les informaría de cuanto aconteciese, y un encantamiento por el cual tenía 4 vidas.

Ahí acaba mi narración. El resto aún está por soñar. Para saber el final,



# CHLOAMCHLA

coge los mandos y reencarnate en Chloamchla.

El juego consta de 7 cámaras las cuales deberás ir pasando. En cada una hay escondido una joya de las que el brujo robó. Los dioses te sacarán automáticamente de allí en cuanto cojas la 7.ª joya.

Por supuesto que no toda la dificultad es esa; el brujo no está dispuesto a dejarse arrebatar las joyas y para impedirte manda monstruos estáticos que te saldrán al paso.

Ten cuidado con las paredes y demás obstáculos.

## Teclas del juego

Arriba ..... A  
 Abajo ..... Z  
 Derecha ..... \  
 Izquierda ..... /

o bien el Joystick.

## «SOLO PARA TRAMPOSOS»

Este apartado va dedicado a aquellos «tocones» que se dedican a urgar por todo el listado para conseguir un éxito fácil y rápido en el juego.

Por ser el primer número le hemos facilitado las trampas y le vamos a proporcionar todo el trabajo hecho.

Con este fin, iremos dando las líneas en las que hay que corregir algo, (con ello ya corregido), para cada tipo de trampas.

### 1. Para tener 1000 vidas

Cambia la línea 440 por esta  
 440 v=1000

### 2. Para empezar en otras cámaras

Cambia:  
 2 empezar en la 2  
 3 empezar en la 3

450 ca = 4 empezar en la 4  
 5 empezar en la 5  
 6 empezar en la 6  
 7 empezar en la 7

## RUTINAS PRINCIPALES

LINEA	FUNCION
10-50	— Identificación del programa.
60-410	— Definición de caracteres, tanto muñecos y decorados como letras.
420	— Llamada a la rutina de portada.
430-530	— Fijación de colores, papeles y ventanas.
540-720	— Desvío: Pantalla inicial: marco del juego. Aparición de marcadores y desvío a las distintas pantallas según corresponda.
730-840	— 1.ª cámara. Tras sacarla a pantalla, llama a bucle principal para que se ejecute el juego.
850-940	— 2.ª cámara.
950-1070	— 3.ª cámara.
1080-1200	— 4.ª cámara.
1210-1330	— 5.ª cámara.
1340-1460	— 6.ª cámara.
1470-1650	— 7.ª cámara.
1660-1960	— Bucle principal: mueve muñeco y examina sus choques. Sirve de contador para la aparición del monstruo. A su vez, desvía a distintas cámaras si en alguna de ellas se ha obtenido la joya.
1970-2020	— Aparición de los monstruos. Puesta a 0 del contador de apariciones. Aumento del oráculo.
2030-2130	— El jugador cuenta con una vida menos. Según la cámara donde se haya producido el choque así manda a una y otra. Si se te han acabado las vidas manda al final.
2140-2280	— Pantalla de felicitación si se ha logrado el objetivo. Exposición de puntos y opción de volver a empezar.
2290-2430	— Pantalla de consolación, si el objetivo no ha sido conseguido. Entrega de puntos y opción de volver a jugar.
2440-2930	— Pantallas de presentación y explicación del juego. Tras instrucciones, elección de nivel y llamada a comienzo de juego.
2940-3080	— Portada inicial. Tras autoejecutarse pide pulsar tecla para inicial instrucciones.

## Fórmulas clave

FORMULA	FUNCION
CA=CA+1	— Significa que el jugador ha alcanzado su objetivo y pasa de cámara.

### 3. Para que no aparezcan monstruos

En la línea 2000 coloca un REM (JUSTO detrás del número). Si te arrepientes quítalo, y sin problemas.

FORMULA	FUNCION
call &BD19	— Es una llamada en código máquina que hace que el movimiento de los personajes sea más suave. Esta fórmula os puede ser muy útil a la hora de hacer vuestros propios programas.
Test (((x-1)*16)+2, (400-(y*16)))	— Nos da como resultado el valor de la tinta que hay en una coordenada gráfica, que gracias a la ecuación está dentro del carácter que movemos. Es muy usada a la hora de detectar choques entre caracteres, de un modo rápido. Esta fórmula sólo vale para el modo 1, para el modo 0 sería: Test (((x-1)*32)+2, (400-(y*16))) Para el modo 2 sería: Test (((x-1)*8)+2, (400-(y*16)))
nk=nk+1:if nk=5+gd...	— Se utiliza como contador. Por cada vuelta que da el bucle principal, que es el que mueve el muñeco, nk aumenta una unidad. Cuando nk alcanza un valor determinado por el nivel de juego, se llama una rutina para que aparezca un monstruo.
o=o+1	— Oráculo = oráculo + 1. Esta fórmula se ejecuta cada vez que aparece un monstruo.
v=v-1	— Significa que el jugador tiene una vida menos. Si v=0 entonces se acaba el juego.
ca*o	— Da los puntos. Es el producto de el número de cámara por los monstruos aparecidos.

#### 6. Para obtener el doble de puntos

Cambia la línea 2440 por:  
2440 PRINT: PEN  
1:PRINT" ";2.ca\*o\*  
Poniendo entre las comillas 12 espacios.

\* Siendo 0 una variable, no un cero.

#### 7. Ultima y nos parece excesiva

Para cambiar su nombre por el de Indiana Jones en las instrucciones. Edita la línea 2890 y varíala a tu conveniencia.

### «Sólo para valientes»

Al igual que en la anterior sección, hemos de decir que no seáis tontos. Aquí tenéis una serie de líneas que os harán el juego más difícil. Así que adelante valientes:

#### 1. Vida única

Cambia la línea 520 por esta  
520 v=1

#### 2. Menos vidas

Cambia la constante de la línea anteriormente citada por cualquier otra que deseéis.

#### 3. Contabilizar menos monstruos

Tan solo teclea la siguiente línea.  
1980 left nk=nk+1:O=INT (0.2+O)

\* donde O es la variable que mide el n.º de monstruos aparecidos.

#### 4. Para apariciones monstruosas casi inmediatas

Teclea:  
1880 nknk+>(FIX gd/5):IF nk>5+gd  
THEN GOSUB 1970

#### 5. Para no acabar nunca

Varía la 2150 por la siguiente.  
2150 LET ca=1:GOTO 540  
Bueno, supongo que ya habrá sido suficiente. Espero que estos cambios os ayuden a comprender la estructura del programa e ir, de este modo, aprendiendo a hacer vuestros propios programas.

Una advertencia: sería bueno que tuviérais una hoja y un lápiz a mano para poder copiar una línea antes de cambiarla por si queréis luego rectificar.

### Variables Principales

VARIABLE	FUNCION
V	— Representa las vidas que le quedan al jugador.
ca	— Indica en que cámara se encuentra el jugador.
L,c,h,m,e,c,H,n k,j,b,d,	— Variables de uso general.
x,y	— Coordenadas del muñeco en la pantalla (mode 1:40x25).
gd	— Selecciona el grado de dificultad que se ha elegido.
nk	— Contador de vueltas del bucle principal. Cuando alcanza un número determinado aparece un monstruo en pantalla.
o	— Número de monstruos que han aparecido. Al final el resultado se calcula multiplicando estos por el n.º de cámara en la que te hayas quedado.

#### 4. Para que cada vez que te mates tengas una vida más

Sustituye la línea 2040 por la siguiente: 2040 v=v+1

#### 5. Vidas infinitas

Si quieres vidas infinitas, cambia la línea 2040 por esta:  
2040 v=v

```

10 '*****
20 '***** CHLOAMCHLA *****
30 '*****
40 '***** frco blazquez *****
50 '***** editorial cometa (c) *****
60 '***** DEF.CARACTERES *****
70 INK 0,0:BORDER 0:INK 1,26:PAPER 0:CLS
80 CLS
90 SYMBOL AFTER 80
100 SYMBOL 98,32,32,44,38,38,44,56,48
110 SYMBOL 99,0,14,24,48,96,48,24,14
120 SYMBOL 100,4,4,52,100,52,28,12
130 SYMBOL 101,0,14,24,48,124,48,24,14
140 SYMBOL 102,24,52,36,32,112,32,32,32
150 SYMBOL 103,0,12,24,48,102,52,28,12
160 SYMBOL 104,2,4,12,24,56,108,198,130
170 SYMBOL 105,0,16,16,16,16,16,16,16
180 SYMBOL 106,4,0,4,4,4,36,44,24
190 SYMBOL 107,0,34,38,44,56,44,38,34
200 SYMBOL 108,32,32,32,32,32,32,112,28
210 SYMBOL 109,&X0,&X100100,&X1011010,&X10000010,&X10000010,&X10000010,&X1000001
0,&X10000010
220 SYMBOL 110,&X0,&X100001,&X1010010,&X10010010,&X10001010,&X10001010,&X1000011
0,&X10000010
230 SYMBOL 95,0,0,32,112,210,140,6,2
240 SYMBOL 94,0,0,4,12,152,208,112,48
250 SYMBOL 111,0,24,52,102,66,102,44,24
260 SYMBOL 112,48,56,44,38,38,44,32,32
270 SYMBOL 113,0,24,52,102,66,106,44,22
280 SYMBOL 114,0,88,108,64,64,64,64,64
290 SYMBOL 115,0,120,68,32,16,8,68,60
300 SYMBOL 116,16,16,124,16,16,16,16,12
310 SYMBOL 117,0,0,66,66,66,102,60,24
320 SYMBOL 118,0,0,130,130,198,108,56,16
330 SYMBOL 120,0,198,68,40,16,40,68,198
340 SYMBOL 121,0,66,34,20,8,16,96,64
350 SYMBOL 122,0,126,2,4,8,16,32,126
360 SYMBOL 97,&X0,&X0,&X1110011,&X11011110,&X11001100,&X11011110,&X1110011,&X0
370 SYMBOL 255,56,124,214,214,254,68,56,238
380 SYMBOL 254,64,228,238,78,76,88,112,255
390 SYMBOL 253,129,153,255,90,255,195,90,255
400 SYMBOL 252,24,124,142,66,199,131,135,126
410 SYMBOL 251,238,68,68,255,1,255,128,255
420 GOTO 2940
430 INK 0,0:BORDER 0:INK 1,9:INK 2,20:INK 3,25
440 CLS
450 PAPER 0
460 WINDOW R1,3,38,2,18
470 WINDOW R2,3,38,20,23
480 FOR c=1 TO 23:PEN 1
490 LOCATE 2,c:PRINT STRING$(38,CHR$(252)):NEXT
500 PAPER R1,0:CLS R1
510 PAPER R2,1:CLS R2
520 v=4
530 ca=1
540 ' *** DESVIO *****
550 FOR l=100 TO 24 STEP-1.8
560 SOUND 1,1,5

```



# CHLOAMCHLA

```
570 NEXT
580 FOR l=24 TO 140 STEP 1.8
590 SOUND 2,1,5
600 NEXT
610 SOUND 5,1000,100
620 PEN R2,3:PAPER R2,1:PRINT R2,"   vidas   oraculo   camara   nivel

"
630 LOCATE 1,25:PRINT " CHLOAMCHLA -ED.COMETA(C)-FCO.BLAZQUEZ"
640 PAPER 0:INK 0,0
650 IF ca=1 THEN GOTO 730
660 IF ca=2 THEN GOTO 850
670 IF ca=3 THEN GOTO 950
680 IF ca=4 THEN GOTO 1080
690 IF ca=5 THEN GOTO 1210
700 IF ca=6 THEN GOTO 1340
710 IF ca=7 THEN GOTO 1470
720 IF ca=8 THEN GOTO 2140
730 '***** 1 CAMARA *****
740 LOCATE 3,6:PEN 1:PRINT STRING$(33,CHR$(254))
750 LOCATE 6,9:PEN 1:PRINT STRING$(33,CHR$(254))
760 FOR H=1 TO 8:LOCATE 20,10+H:PEN 1:PRINT CHR$(254):NEXT
770 FOR H=1 TO 8:LOCATE 26,8+H:PEN 1:PRINT CHR$(254):NEXT
780 FOR H=1 TO 8:LOCATE 32,10+H:PEN 1:PRINT CHR$(254):NEXT
790 LOCATE 33,15:PEN 2:PRINT STRING$(4,CHR$(251))
800 FOR h=1 TO 10 STEP 2:LOCATE 4+h,9+h:PEN 2:PRINT CHR$(252):NEXT
810 x=4:y=4
820 LOCATE 33,18:PEN 3:PRINT CHR$(180)
830 IF x=33 AND y=18 THEN ca=ca+1:IF ca=2 GOTO 540
840 GOTO 1660
850 '***** 2 CAMARA *****
860 CLS R1
870 FOR n=1 TO 12:LOCATE 2+n,3+n:PEN 2:PRINT STRING$(20,CHR$(251))
880 LOCATE 2+n,12+n/2:PEN 2:PRINT CHR$(251)
890 LOCATE 24+n,2+n:PEN 1:PRINT CHR$(254)
900 LOCATE 16+(n+n/1.5),18:PEN 3:PRINT CHR$(254):NEXT
910 x=3:y=5
920 LOCATE 3,2:PEN 3:PRINT CHR$(180)
930 IF x=3 AND y=2 THEN ca=ca+1:IF ca=3 THEN GOTO 540
940 GOTO 1660
950 '***** 3 CAMARA *****
960 CLS R1
970 FOR m=1 TO 16 STEP 2:LOCATE 7,2+m:PEN 2:PRINT STRING$(10+m,CHR$(252))
980 LOCATE 27+(m/2),3+m:PEN 1:PRINT CHR$(254)
990 LOCATE 5,1+m:PEN 1:PRINT CHR$(254)
1000 LOCATE 4,2+m:PEN 2:PRINT CHR$(251):NEXT
1010 LOCATE 23,10:PEN 1:PRINT CHR$(253)
1020 FOR p=1 TO 2:LOCATE 5+p,17:PEN 1:PRINT CHR$(254):NEXT
1030 LOCATE 5,6:PRINT " "
1040 x=37:y=17
1050 LOCATE 22,10:PEN 3:PRINT CHR$(180)
1060 IF x=22 AND y=10 THEN ca=ca+1:IF ca=4 THEN GOTO 540
1070 GOTO 1660
1080 '***** 4 CAMARA *****
1090 CLS R1
1100 FOR e=1 TO 16 STEP 1.5:LOCATE 5,2+e:PEN 2:PRINT STRING$(29,CHR$(251)):NEXT
1110 FOR e=1 TO 16 STEP 2:LOCATE 20,2+e:PRINT " ":NEXT
```

```

1120 FOR e=1 TO 16 STEP 2:LOCATE 7,1+e:PRINT " ":NEXT
1130 FOR e=1 TO 16 STEP 5.5:LOCATE 34,2+e:PEN 1:PRINT STRING$(5,CHR$(254)):NEXT
1140 FOR e=1 TO 16 STEP 2:LOCATE 30,1+e:PRINT " ":NEXT
1150 FOR e=1 TO 16 STEP 5.5:LOCATE 3,5+e:PEN 1:PRINT STRING$(5,CHR$(254)):NEXT
1160 LOCATE 5,2:PEN 3:PRINT CHR$(180);" "
1170 LOCATE 22,18:PRINT " "
1180 x=22:y=18
1190 IF x=5 AND y=2 THEN ca=ca+1:IF ca=5 THEN GOTO 540
1200 GOTO 1660
1210 ' ***** 5 CAMARA *****
1220 CLS R1
1230 FOR c=1 TO 16:LOCATE 9,1+c:PEN 1:PRINT CHR$(251):NEXT
1240 FOR c=1 TO 14:LOCATE 6,4+c:PEN 1:PRINT CHR$(251):NEXT
1250 FOR h=5 TO 20 STEP 6:PEN 1:LOCATE 9,h:PRINT STRING$(27,CHR$(254)):NEXT
1260 FOR h=5 TO 16 STEP 6:PEN 1:LOCATE 12,h+3:PRINT STRING$(27,CHR$(254)):NEXT
1270 LOCATE 4,18:PEN 3:PRINT CHR$(180)
1280 x=11:y=4
1290 FOR n=3 TO 16 STEP 3:LOCATE 20,n+1:PEN 2:PRINT CHR$(253)
1300 LOCATE 30,n+1:PEN 2:PRINT CHR$(253):NEXT
1310 LOCATE 30,17:PRINT " "
1320 IF x=4 AND y=18 THEN ca=ca+1:IF ca=6 THEN GOTO 540
1330 GOTO 1660
1340 ' ***** 6 CAMARA *****
1350 CLS R1
1360 FOR n=7 TO 18:PEN 1:LOCATE 3,n:PRINT STRING$(36,CHR$(251)):NEXT
1370 FOR n=6 TO 18 STEP 2.5:LOCATE 4,n:PRINT "
":NEXT
1380 FOR n=6 TO 16 :LOCATE 10,n:PRINT " ":NEXT
1390 FOR n=6 TO 16 :LOCATE 30,n:PRINT " ":NEXT
1400 FOR k=3 TO 38 STEP 2.5:LOCATE k,4:PEN 2:PRINT CHR$(254):NEXT
1410 FOR k=3 TO 38 STEP 1.5:LOCATE k,6:PEN 2:PRINT CHR$(254):NEXT
1420 FOR k=3 TO 38 STEP 1.05:LOCATE k,3:PEN 2:PRINT CHR$(254):NEXT
1430 LOCATE 4,16:PEN 3:PRINT CHR$(180)
1440 x=21:y=2
1450 IF x=4 AND y=16 THEN ca=ca+1:IF ca=7 THEN GOTO 540
1460 GOTO 1660
1470 ' ***** 7 CAMARA *****
1480 CLS R1
1490 LOCATE 4,3:PEN 2:PRINT STRING$(35,CHR$(252))
1500 LOCATE 6,5:PEN 3:PRINT STRING$(33,CHR$(251))
1510 LOCATE 6,16:PEN 3:PRINT STRING$(30,CHR$(251))
1520 FOR j=1 TO 11:LOCATE 36,6+j:PEN 3:PRINT CHR$(251)
1530 LOCATE 6,5+j:PEN 3:PRINT CHR$(251):NEXT
1540 LOCATE 7,8:PEN 1:PRINT STRING$(27,CHR$(254))
1550 LOCATE 7,13:PEN 1:PRINT STRING$(27,CHR$(254))
1550 LOCATE 7,13:PEN 1:PRINT STRING$(27,CHR$(254))
1560 FOR b=8 TO 13:PEN 1:LOCATE 7,b:PRINT CHR$(254)
1570 LOCATE 33,b:PRINT CHR$(254):NEXT
1580 LOCATE 9,10:PEN 3:PRINT STRING$(22,CHR$(253))
1590 LOCATE 9,11:PEN 2:PRINT STRING$(22,CHR$(252))
1600 FOR l=9 TO 31 STEP 2:LOCATE l,10:PRINT " ":NEXT
1610 LOCATE 9,13:PRINT " "
1620 LOCATE 38,2:PEN 3:PRINT CHR$(180)
1630 x=21:y=10
1640 IF x=38 AND y=2 THEN ca=ca+1:IF ca=8 THEN GOTO 540
1650 GOTO 1660
1660 ' ***** BLOQUE COMUN *****
1670 LOCATE 7,22:PAPER 1:PEN 3:PRINT v
1680 LOCATE 26,22:PAPER 1:PEN 3:PRINT ca

```

# CHLOAMCHLA

```
1690 LOCATE 34,22:PAPER 1:PEN 3:PRINT gd
1700 PAPER 0
1710 PEN 3
1720 IF NOT TEST((((x-1)*16)+2),(400-(y*16)))=0 THEN GOTO 2030
1730 LOCATE x,y:CALL &BD19:PRINT CHR$(255)
1740 CALL &BD19
1750 LOCATE x,y:CALL &BD19:PRINT " "
1760 IF INKEY(73)=0 THEN y=y+1
1770 IF INKEY(71)=0 THEN y=y+1
1780 IF INKEY(72)=0 THEN y=y-1
1790 IF INKEY(69)=0 THEN y=y-1
1800 IF INKEY(75)=0 THEN x=x+1
1810 IF INKEY(22)=0 THEN x=x+1
1820 IF INKEY(74)=0 THEN x=x-1
1830 IF INKEY(30)=0 THEN x=x-1
1840 IF x<3 THEN x=3
1850 IF x>38 THEN x=38
1860 IF y<2 THEN y=2
1870 IF y>18 THEN y=18
1880 nk=nk+1:IF nk=5+gd THEN GOSUB 1970
1890 IF ca=1 THEN GOTO 830
1900 IF ca=2 THEN GOTO 930
1910 IF ca=3 THEN GOTO 1060
1920 IF ca=4 THEN GOTO 1190
1930 IF ca=5 THEN GOTO 1320
1940 IF ca=6 THEN GOTO 1450
1950 IF ca=7 THEN GOTO 1640
1960 IF ca=8 THEN GOTO 2140
1970 '**** EVERY *****
1980 LET nk=0:0=0+1
1990 LOCATE 16,22:PAPER 1:PEN 3:PRINT 0
2000 LOCATE x+((RND(1)*2-RND(1)*2)),y+((RND(1)*2-RND(1)*1)):PAPER 0:PEN 2:PRINT
CHR$(253):SOUND 2,60,20
2010 LOCATE x,y:PAPER 0:PRINT " "
2020 RETURN
2030 '***** MUERTE *****
2040 v=v-1
2050 SOUND 3,600,30
2060 IF v=0 THEN GOTO 2290
2070 IF ca=1 THEN GOTO 730
2080 IF ca=2 THEN GOTO 850
2090 IF ca=3 THEN GOTO 950
2100 IF ca=4 THEN GOTO 1080
2110 IF ca=5 THEN GOTO 1210
2120 IF ca=6 THEN GOTO 1340
2130 IF ca=7 THEN GOTO 1470
2140 ' ***** VENCEDOR *****
2150 CLS
2160 LOCATE 4,4:PEN 2:PRINT " los dioses son contigo !!!!"
2170 PRINT
2180 PRINT
2190 PEN 3:PRINT " toda mittokropolis esta en pie
esperando tu regreso:eres un her
oe"
2200 PRINT
2210 PRINT " tu nombre sera leyenda en el libro de
los dioses"
```

```

2220 PRINT
2230 PRINT "      eres el salvador del olimpo.. has
      sido recompensado con .....la vida
      eterna ademas de ...."
2240 PRINT:PEN 1:PRINT "      ";ca%o
2250 PEN 3:PRINT:PRINT "      favores divinos"
2260 LOCATE 6,25:PEN 1:PRINT "pulsatecla <z>para empezar"
2270 IF INKEY(71)=0 THEN GOTO 2440
2280 GOTO 2270
2290 ' ***** PERDEDOR *****
2300 CLS
2310 LOCATE 4,4:PEN 2:PRINT "      mala suerte !!!!!!"
2320 PRINT
2330 PRINT
2340 PEN 3:PRINT "      en mittokropolis todo el mundo
      llora tu derrota..."
2350 PRINT
2360 PRINT " no te preocupes..te haran un monumento...."
2370 PRINT
2380 PRINT "      en cualquier caso,los dioses
      saben de la valia de tus proezas
      gracias al oraculo.....te conceden ..."

2390 PEN 1:PRINT:PRINT "      ";ca%o
2400 PEN 3:PRINT "      favores divin
os"
2410 LOCATE 6,25:PEN 1:PRINT "pulsatecla <z>para empezar"
2420 IF INKEY(71)=0 THEN GOTO 2440
2430 GOTO 2420
2440 ' ***** PRESENTACION *****
2450 CLEAR
2460 MODE 1:INK 0,0:BORDER 0:PAPER 0
2470 LOCATE 2,6:PEN 3:PRINT " se atreve ud. a vivir una anentura
      en una antigua civiliz
acion perdida?"
2480 PRINT
2490 PRINT
2500 PRINT
2510 PEN 2:PRINT "      venga entonces a mittokropolis      "
2520 PRINT
2530 LOCATE 10,24:PEN 1:PRINT " pulsatecla <z> para comenzar "
2540 IF INKEY(71)=0 THEN GOTO 2560
2550 GOTO 2540
2560 CLS
2570 LOCATE 15,4:PEN 2:PRINT "CHLOAMCHLA"
2580 LOCATE 14,5:PEN 1:PRINT "-----"
2590 LOCATE 3,7:PEN 3:PRINT " chloamchla,ciudadano de mittokropolis,
      es elegido por los dios
es para recuperar      las 7 joyas magica
s que el brujo      "
2600 PRINT " kenthonn ha robado.."
2610 PRINT
2620 PRINT "      por avatares del destino(y por querer
      jugar),tu te has reencarnado en
      "
2630 PRINT "      chloamchla y debes llevar a cabo
      tu mision.. "
2640 PRINT

```

# CHLOAMCHLA

```
2650 PRINT " para ello cuentas con 4 vidas "
2660 PRINT:PRINT:PEN 1:PRINT " pulsa <z> para seguir "
2670 IF INKEY(71)=0 THEN GOTO 2690
2680 GOTO 2670
2690 CLS
2700 LOCATE 6,4:PEN 3:PRINT " has viajado hasta los dominios kenthonn
                        estas rodeado de peli
gros y trampas.                                sobretodo ten cuidado c
on los monstruos                                "
2710 PRINT "estaticos que kenthonn te envia continua
           -mente"
2720 PRINT:PRINT " con habilidad podras usar su propios
           hechizos para eliminar obstaculos"
:PRINT
2730 PRINT " hay varios tipos de trampas:"
2740 PRINT
2750 PEN 1:PRINT " ";CHR$(254)" :planta venenosa"
2760 PRINT " ";CHR$(253);" :monstruo estatico":PRINT " ";CHR$(
252)" :piedra encantada"
2770 PRINT " ";CHR$(251)" :muralla maldita "
2780 PRINT
2790 PRINT
2800 PEN 2:PRINT " no los olvide..pulse <z> para seguir "
2810 IF INKEY(71)=0 THEN GOTO 2830
2820 GOTO 2810
2830 CLS
2840 LOCATE 3,3:PEN 3:PRINT "Puede jugar con joystick o teclado"
2850 PRINT
2860 PRINT " las teclas son:                                a....arriba
                        z....abajo
                        /....izquierda"
2870 PRINT " ñ....drcha"
2880 PRINT
2890 PRINT " Que nivel quieres :                                45.....torpes
                        38.....pschhh
                        30.....respetable                                15.....heroico
                        10.....indiana jones y cia. "
2900 PRINT
2910 INPUT " puedes elegir nivel intermedio";gd:gd=FIX(gd)
2920 IF gd<10 OR gd>45 THEN GOTO 2890
2930 GOTO 430
2940 ' ***** PORTADA *****
2950 CLS
2960 INK 0,0:BORDER 0:INK 1,9,12:INK 2,20:INK 3,25
2970 FOR d=1 TO 10 STEP 2:LOCATE 37-d,21:PEN 3:PRINT CHR$(252):NEXT
2980 LOCATE 2,22:PEN 1:PRINT STRING$(36,CHR$(254))
2990 FOR e=1 TO 20 STEP 2:PEN 2:LOCATE 2,2+e:PRINT STRING$(e,CHR$(251))
3000 PEN 3:LOCATE 1+e,1+e:PRINT CHR$(253):NEXT
3010 LOCATE 24,21:PEN 2:PRINT CHR$(255)
3020 FOR h=185 TO 538 STEP 2:DEG:ORIGIN 400,275:PLOT 115*SIN(h),35*COS(h),3:NEXT
3030 ORIGIN 390,240:DRAW -10,-140,3
3040 ORIGIN 405,240:DRAW -25,-140,3
3050 LOCATE 20,8:PEN 1:PRINT "chloamchla!"
3060 LOCATE 1,23:PEN 2:PRINT " PULSA TECLA..."
3070 WHILE INKEY$="":WEND
3080 GOTO 2440
```

**S**u simple estructura nos brinda la posibilidad de aprender todo sobre él. El listado del mismo, con sus distintas rutinas bien diferenciadas gracias a los REM, nos facilitan aún más esta labor.

Ahora, y apoyándonos en esto, vamos a ir estudiando la estructura del listado, explicando lo necesario y comentado la función y valor, así como otras cualidades de cada rutina.

#### Línea 70

Es quizás la más importante. Dimensiona las matrices de datos que vamos a usar. Sin esta línea sólo podrían ejecutarse correctamente las demás rutinas para números bajos (0-10 aproximadamente). Nosotros hemos dimensionado estas en 15, para que coincidan con cada uno de los colores del modo 0, pero si quieres usar el programa para matrices mayores, cambia las líneas siguientes por las antiguas:

```
820 ORIGIN 33+(n-1) * (570/nba),
25+d:DRAW (570/nba),0,1
```

```
280 PRINT: PRINT: PRINT: "Introduce n.º de bloques a usar": INPUT"";nba
```

```
290 IF nba<2 THEN GOTO 280
```

```
440 LOCATE 1,4+n: PEN 1:
PRINT chr$(143): LOCATE 2,4+n:
PRINT chr$(n): LOCATE 7,4+n:
PRINT ROUND ((v(n)*100/mv), 1):
LOCATE 12,4+n: PRINT v(n):
NEXT
```

```
490 FOR n=1 TO x (Sustituye x por la dimensión de las nuevas matrices)
```

```
630 FOR n=1 TO x (Sustituye x por la dimensión de las nuevas matrices)
```

```
70 DIM va(x): DIM v(x): DIM c$(x)
```

No olvides que cuando teclees este programa no has de poner x, sino una constante numérica; que viene a coincidir con el número de datos que quieres meter.

#### Líneas 80-120

Es la rutina del «Menú principal». Tras escribir en pantalla las opciones (líneas 90-120), el programa entra en un bucle cerrado del cual sólo podrá salir si se elige una de las opciones (líneas 120-210). Dependiendo de tu respuesta así te mandará a una línea o a otra.

**B**arrotos II es un programa ideado para estudios sobre tablas de valores.



Está capacitado para representar gráficamente una tabla completa de valores, para guardar estos en cinta o disco, para cargarlos (cinta o disco), para comprobar cada dato con el máximo valor introducido, etc.

## BARROTOS II

#### Líneas 220-380

Primero se fijan colores. Se indica que estamos en la opción de «introducción de información» (línea 240) y comienza una serie de preguntas que nos facilitan la entrada de datos. Primero te pide el nombre de la información. Mete el que quieras. Tras ello, te pide el máximo valor, es decir, o bien mete de todos los datos el más elevado que tengas, o bien uno no muy mayor. En caso de ser el máximo valor menor que cualquier dato, este último se saldría de la pantalla a la hora de la representación gráfica. Por ello, el valor 0 no te lo admite como máximo valor.

Inmediatamente después, una de las opciones más útiles de este programa. Te pide el número de barras o datos a introducir. Dadas las dimensiones que hemos establecido, el número de éstas variará de 0 a 15 barras, pero



no acepta menos de 2 bloques, pues una gráfica de 0 a 1 dato es totalmente absurda.

Esta opción te permite que a la hora de la representación, la gráfica quede estéticamente bien, y no suceda lo que pasaría si fuera un programa de bloques fijos. Por ello, en la gráfica ocupan igual a lo ancho 2 barras que 15. Elige según tu necesidad.

Habiendo seleccionado ya el n.º de bloques, deberemos introducir el concepto y el valor de cada uno. Si no quieres introducir el concepto o el valor de algún bloque, basta con que des al ENTER cuando te lo pida.

Y por último en esta rutina, te ofrece la posibilidad de rectificar la entrada de datos. Para ello responde r o R y volverá a comenzar el interrogatorio.

La línea más importante de esta rutina es la 340. Fíjate en ella.

#### Líneas 390-450

La opción de «Pizarra de Información». En el modo 0 y tantos colores como barras hayas elegido, aparece en pantalla el concepto, el tanto por ciento relativo y el valor real de cada bloque. El color de esta información coincide con su barra en la gráfica. OBSERVESE LA LINEA 440, donde se haya el % de cada valor con el máximo valor.

#### Líneas 460-590

Con esta opción podrás cargar desde cinta o disco cualquier tabla de valores. Al preguntarte el nombre del fichero a cargar no puedes no introducir nada pues te repite la pregunta. Esto se debe a que si estás usando una unidad de disco y no metes ningún nombre, daría error. Para evitar esto, dale el nombre del fichero deseado y espera a que se cargue.

#### Línea 600-730

Esta rutina te permite grabar en un disco o cinta los datos ya introducidos. Para ello, introduce el nombre que desees y pulsa REC y PLAY. Por los mismos motivos de antes no te admite nombres en blanco. También por exigencias del disco, tu nombre no será mayor de 6 letras, y si lo es, quedará grabado sólo como si fueran las 6 primeras.

FORMULAS CLAVE	
FORMULA	FUNCION
$V(a)=((v(n)*350)/mv)$	— Convierte el valor real de cada uno de los datos, a valores a escala del máximo valor introducido, de modo que su ajuste en pantalla sea perfecto.
ROUND $((v(n)*100/mv),1)$	— Escribe el tanto por ciento respecto al máximo valor que representa el valor (n). El round le obliga a dar tanto sólo 1 decimal.
6\$-LETF\$(c\$(n),5)	— Convierte la cadena caracteres literales c\$ en tan solo sus cinco primeros caracteres comenzando por la izquierda.
ORIGIN $33+(n-1)*(570/nba),$ 25+d	— Tomando en cuenta el máximo valor, asigna un origen a cada columna. El segundo término (25+d) fija la altura de la barra según el valor de d.
DRAW $(570/nba),0,n$	— Calcula el valor para cada columna según el máximo valor. El tercer término, (n), fija el color que va a llevar esa barra.

VARIABLES PRINCIPALES	
VARIABLE	FUNCION
DIM va(15)	— Dimensionada una matriz de 15 elementos, pertenecientes a la variable va.
DIM v(15)	— Al igual que arriba, pero para la variable v.
DIM c\$(15)	— La misma función que las anteriores.
ndi\$	— Nombre de la información
mv	— Máximo valor, tomado como referencia para poner a escala de pantalla todos los datos a introducir.
nba	— Número de bloques que se van a usar. Sirve para determinar el grosor de cada barra según el número elegido.
c\$(n)	— Conceptos de los distintos bloques a usar.
v(n)	— Valor real de cada una de las barras o datos.
va(n)	— Valor convertido de cada uno de los datos o barras.
r\$	— Variable usada para rectificar datos si r\$="r".
nf\$	— Nombre del fichero a cargar o grabar.
n, m,	— Variable de uso general para bucles y sucesiones
d	— Compone la escala de valores de 1 a v(n) para que se dibujen las barras.

Ambos ajustes los podrás estudiar en las líneas 470 y 610.

#### Líneas 740-760

«Fin de programa». Sales del programa y las tintas vuelven a su color inicial. El programa aún está en memoria. Teclea list para listarlo o bien NEW si es que quieres cargar otro

programa.

#### Líneas 770-860

Es la representación gráfica de los valores de la tabla por ti introducidos. Fíjate en las fórmulas de la línea 820, en las que se determina el origen de cada barra, su grosor, su altura y su color. Son las más importantes de esta rutina.




FIGURA III

```

10 ' *****
20 '*****  BARROTES II *****
30 '*****          *****
40 '***** FCO.J.BLAZQUEZ *****
50 '***** ED. COMETA (c) *****          ***** 1985 *****
60 '*****
70 DIM va(15):DIM v(15):DIM c$(15)
80 '----- menu principal -----
90 MODE 1:INK 0,0:BORDER 0:INK 1,26:INK 2,11:PAPER 0:LOCATE 1,25:PEN 3:PRINT "
   Fco.J.BLAZQUEZ (c) ED.COMETA-1985":LOCATE 1,1
100 PRINT:PRINT:PRINT:PEN 2:PRINT "          MENU PRINCIPAL
      ----- "
110 PRINT:PEN 1:PRINT:PRINT "          1.... INTRODUCIR INFORMACION.
      2.... VER GRAF
ICA.
RRA."          3.... VER PIZA
120 PRINT:PRINT "          4.... LOAD INFORMACION.
          5.... SAVE INFORMACION.
          6.... FINALIZAR PROGRAMA"
130 a$=INKEY$
140 IF a$="" THEN GOTO 130
150 IF a$="1" THEN GOTO 220
160 IF a$="2" THEN GOTO 770
170 IF a$="3" THEN GOTO 390
180 IF a$="4" THEN GOTO 460
190 IF a$="5" THEN GOTO 600
200 IF a$="6" THEN GOTO 740
210 GOTO 130
220 '--- Introducir informacion ----
230 MODE 1:INK 1,26:INK 0,0:BORDER 0:INK 3,12
240 PRINT:PRINT:PRINT:PEN 3:PAPER 1:PRINT "          ENTRADA DE INFORMACION
      ":PAPER 0:PEN 2
250 PRINT:PRINT:PRINT:PRINT " Introduce nombre de la informacion ":INPUT " ";n
di$
260 PRINT:PRINT:PRINT " Introduce Maximo valor ":INPUT " ";mv
270 IF mv=0 THEN PRINT " No admito cifras tan cercanas a 0 ":GOTO 260
280 PRINT:PRINT:PRINT " Introduce numero de bloques a usar ":INPUT " (2-15) ";nb
a
290 IF nba<2 OR nba>15 THEN GOTO 280
300 FOR n=1 TO nba
310 PRINT:PRINT " Introduce concepto del bloque n:":n:INPUT " ";c$(n)
320 LET c$(n)=LEFT$(c$(n),5)
330 PRINT:PRINT " Introduce valor de ";c$(n):INPUT " ";v(n)
340 LET va(n)=((v(n)*350)/mv)
350 NEXT
360 PRINT:PRINT:PRINT " Para rectificar pulsa << r >> ":INPUT " ",r$
370 r$=UPPER$(r$):IF r$="R" THEN GOTO 220
380 GOTO 80
390 '**** PIZARRA DE INFORMACION ****
400 MODE 0
410 MODE 0:INK 0,0:BORDER 0:INK 1,12:INK 2,13:INK 3,15:INK 4,19:INK 5,1:INK 6,26
:INK 7,7:INK 8,8:INK 9,3:INK 10,24::INK 11,6:INK 12,
11:INK 13,5:INK 14,15:INK 15,23
420 PRINT:PEN 3:PRINT " CONC. %          VALOR _____":PRINT
430 PRINT:FOR n=1 TO nba
440 LOCATE 1,4+n:PEN n:PRINT CHR$(143):LOCATE 2,4+n:PRINT c$(n):LOCATE 7,4+n:PRI
NT ROUND ((v(n)*100/mv),1):LOCATE 12,4+n:PRINT v(n):
NEXT
450 LOCATE 1,25:PEN 9:PRINT "          PULSA TECLA_":WHILE INKEY$="":WEND:GOTO 80
460 '***** LOAD INFORMACION *****
470 CLS:MODE 1:INK 1,2:PEN 1:INK 2,26:PAPER 2:PRINT:PRINT:INPUT "          INTRODUCE NOM

```

# BARROTES II

```
BRE DE FICHERO ";nf$:nf$=LEFT$(nf$,6):PRINT:PRINT:IF
nf$="" THEN PRINT " HAS DE INTRODUCIR UN NOMBRE ":FOR m=1 TO 2000:NEXT:GOTO 46
0
480 OPENIN nf$
490 FOR n=1 TO 15
500 INPUT R9,nba
510 INPUT R9,v(n)
520 INPUT R9,va(n)
530 INPUT R9,c$(n)
540 INPUT R9,mv
550 INPUT R9,ndi$
560 NEXT
570 CLOSEIN
580 PRINT:PRINT:PAPER 0:PEN 2:PRINT " FICHERO CARGADO !!!!!!!"
590 FOR m=1 TO 3000:NEXT:SOUND 1,300,20:CLS:GOTO 80
600 '***** SAVE INFORMACION *****
610 CLS:MODE 1:INK 1,2:PEN 1:INK 2,26:PAPER 2:PRINT:PRINT:INPUT " INTRODUCE NOM
BRE DE FICHERO ";nf$:nf$=LEFT$(nf$,6):PRINT:PRINT:IF
nf$="" THEN PRINT " HAS DE INTRODUCIR UN NOMBRE ":FOR m=1 TO 2000:NEXT:GOTO 60
0
620 OPENOUT nf$
630 FOR n=1 TO 15
640 PRINT R9,nba
650 PRINT R9,v(n)
660 PRINT R9,va(n)
670 PRINT R9,c$(n)
```

## Boletín de Suscripción

Remitir a Editorial Cometa, s.a. C/ Zamora s/n nave 11. San Fernando de Henares (MADRID)

Deseo suscribirme a los 11 números anuales de P. mi AMSTRAD por sólo 3.500 pts. (Vd. ahorra 1.450 pts.)

El importe lo haré efectivo:

- Por giro postal n.º .....
- Por talón nominativo adjunto.
- Contra reembolso a la recepción del primer ejemplar, más gastos de envío.

Deseo suscribirme a partir del n.º ..... (inclusive).

Nombre y apellidos: ..... Telef. ....

Domicilio: ..... Firma

Ciudad: .....

Provincia .....

# BARROTES II

```
680 PRINT R9,mv
690 PRINT R9,ndi$
700 NEXT
710 CLOSEOUT
720 PRINT:PRINT:PAPER 0:PEN 2:PRINT " FICHERO GRABADO !!!!!!!"
730 FOR m=1 TO 2000:NEXT:SOUND 1,300,20:CLS:GOTO 80
740 '***** FINALIZAR *****
750 CLS:INK 0,2:BORDER 2:INK 1,24:PEN 1
760 END
770 '***** GRAFICA *****
780 MODE 0:INK 0,0:BORDER 0:INK 1,12:INK 2,13:INK 3,15:INK 4,19:INK 5,1:INK 6,26
:INK 7,7:INK 8,8:INK 9,3:INK 10,24::INK 11,6:INK 12,
11:INK 13,5:INK 14,15:INK 15,23
790 ORIGIN 30,22:DRAW 580,0,6:DRAWR 0,356:DRAWR -580,0:DRAWR 0,-356
800 FOR n=1 TO nba
810 FOR d=1 TO va(n) STEP 2
820 ORIGIN 33+(n-1)*(570/nba),25+d:DRAW (570/nba),0,n
830 NEXT d:NEXT n
840 LOCATE 1,25:PRINT " PULSA TECLA "
850 WHILE INKEY#="":WEND
860 GOTO 80
```



ZAMORA, s/n. Nave 11 San Fernando de Henares (MADRID)



# LO ÚNICO...

# UN CHORRO EFICAZ

PARA UNA BUENA CARGA  
DEL PROGRAMA EN TU ORDE-  
NADOR, COMPRUEBE  
QUE LA CABEZA DE SU  
PLATINA ESTE  
LIMPIA Y EL  
AZIMUTH PERFECTA-  
MENTE AJUSTADO.  
LIMPIELA CON EL  
PRODUCTO  
SKUA.



## LIMPIADOR DE RADIO CASSETTES



FABRICADO POR

### ASKLE CHEMICAL S.A.

ALICANTE (SPAIN)

### BOLETIN DE PEDIDO

Nombre y Apellidos .....  
Direccion .....  
Ciudad ..... Telf. .... D.P. ....

Deseo recibir..... Limpiadores de cassettes al precio de 420 Ptas. Unidad.  
Forma de pago:  Talón  Contrareembolso  Giro postal

C/. Zamora S/N. nave 11, San Fernando de Henares (MADRID)

# AMPLIA LAS POSIBILIDADES DE TU AMSTRAD



**IMPRESORA PRINTER 130**  
Especialmente recomendada para ordenadores AMSTRAD. **54.900 Pts**



**JOYSTICKS**  
Los famosos SVI de la serie Quickshot. Desde **1.600 Pts**



**LAPIZ OPTICO**  
Diseña gráficos y menús de comunicación en la pantalla a color. Incluye software. **4.500 Pts**



**INTERFACE SERIE RS 232 C**  
Para conectar con modems, impresoras serie u otros ordenadores. **11.750 Pts.**



**UNIDAD DE DISCO**  
Incluye Sistema Operativo CP/M y lenguaje LOGO. **45.500 Pts** (con controlador). **39.500 Pts** (sin controlador).



**SINTETIZADOR DE VOZ**  
Emula la voz humana. Incluye dos altavoces y el software. **9.000 Pts**



**MODULADOR TV COLOR**  
Para utilizar el TV como pantalla a todo color. **9.000 Pts** (CPC 464), **9.450 Pts** (CPC 664 y 6128)

**AMSTRAD**

ESPAÑA

GRUPO INDESCOMP

Avda. del Mediterráneo, 9. Tels. 433 45 48 - 433 48 76 - 28007 MADRID

Delegación Cataluña: Tarragona, 110 - Tel. 325 10 58. 08015 BARCELONA